

Doc No.: AN-270

Version: 2.5

Date: 07 February 2013

Subject: MC464 Panasonic RTEX

APPLICATION NOTE

www.triomotion.com

1. Introduction:

This technical note provides notes to explain some of the MC464 Panasonic RTEX interface features which are not yet in the Technical Reference Manual. The document also describes how to use the new features of the A5N drives: speed and torque modes, registration, and the expanded parameter lists.

2. Applicability:

Information applies to MC464 with RTEX32 interface for Panasonic A4N/A5N digital drives.

MC464 Software Version 2.0153 and later.

If the Motion Coordinator SERVO_PERIOD is 500 μ sec the RTEX communication cycle should also be set to 500 μ sec using Panaterm.

The A5N default servo cycle time is 500 μ sec, so the MC464 must be set to SERVO_PERIOD=500 or the drives must be changed to 1000 μ sec.

Panasonic A5N Drive Parameters		
SERVO_PERIOD	7.20 : RTEX communication cycle setup.	7.21 : RTEX command updating cycle ratio.
500 μ sec	3	2 or 1
1000 μ sec	6	1

If the drive is set to 500 μ sec and the Motion Coordinator to 1000 μ sec it is NOT possible to change the drive parameters via the controller.

3. Initialisation:

During controller initialisation or software reset the RTEX bus is scanned for drives. The MC464 sets an axis parameter DRIVE_TYPE to 4 for an A4N drive and 5 for an A5N drive.

Axes are allocated to the drive ADDRESS switch setting + 1. So for example a drive set to address 2 will be allocated to axis 3 in the controller.

Axes are set by default to ATYPE=50. The ATYPE can be assigned to 51 or 52 to select Speed or Torque modes on A5N drives. Profile position mode is not supported.

Drive Mode:	ATYPE
Cyclic Position	50
Cyclic Speed	51
Cyclic Torque	52

4. Drive Parameter Commands:

The parameter numbers for A5N and A4N drives are NOT compatible. On the A5N drive parameters have a “Class” number plus a number within the class. The parameter number for the DRIVE_READ/DRIVE_WRITE commands are constructed by multiplying the class number by 256 and adding the number within the class.

For example:

DRIVE_READ(4 * 256 + 16) AXIS(2) ‘ read parameter 16, class 4

```
value = DRIVE_READ(param)
result = DRIVE_READ(param,VR#)
```

DRIVE_READ is an axis function which reads a drive parameter. There are 2 forms of the command. The single parameter form reads a drive parameter and returns the result. This is a quick way to read a drive parameter. For example:

```
>>? DRIVE_READ($19) AXIS(17)
35.0000
```

The two parameter form of the command places the parameter read into a VR value and returns the - 1 for a successful read, 0 for an error during reading. This form is the best to use when putting the DRIVE_READ command into a program:

```
PRINT#5, ""
PRINT#5, "Drive Parameters Axis:";ax[2,0]
PRINT#5, ""
IF DRIVE_TYPE AXIS(ax)=4 THEN

  FOR p= 1 TO 127
    IF DRIVE_READ(p,1) AXIS(ax) THEN
      PRINT#5, HEX(p);":":VR(1)[8,0]
    ELSE
```

```

        PRINT#5, "Error reading parameter: ";HEX(p)
    ENDIF
NEXT p
ELSE
FOR class=0 TO 8
FOR p= 0 TO 43
IF DRIVE_READ(p + (class * 256),1) AXIS(ax) THEN
PRINT#5, class[3,0];":":p[3,0];":":VR(1)[8,0]
ELSE
PRINT#5, "Error reading parameter: ";HEX(p)
ENDIF
NEXT p
NEXT class
ENDIF
NEXT p

```

Note that the Panasonic drive parameters are numbered 1..127. In addition the DRIVE_READ function can be used to access the drive “System ID”, “Alarm” and “Monitor” values:

Drive Function:	Command Format:	Notes:
System ID	DRIVE_READ(\$10XXX)	System ID's are strings. DRIVE_READ command will print string to command line or place into VR variables if the DRIVE_READ(val,vr) format is used XXX is the Panasonic code for the System ID string. For example DRIVE_READ(\$10140) returns the firmware version for the drive.
Alarm Read	DRIVE_READ(\$2Z0XX)	Bits 12..15 of command value (Z) hold the “index” of the alarm number. For example: DRIVE_READ(\$26000,1) places alarm 6 into VR 1 DRIVE_READ(\$20001) performs Clear Alarm function
Monitor Command	DRIVE_READ(\$3ZXXX)	For example DRIVE_READ(\$30201) reads Mechanical Angle, the position within one turn. Note that this is -1 until encoder zero has been seen. DRIVE_READ(\$30102) reads Encoder Resolution.

```
result = DRIVE_WRITE(param,value)
```

DRIVE_WRITE is an axis function which writes a drive parameter. The function always takes 2 parameters and returns the success/fail result. For example:

```
>>> DRIVE_WRITE($19,50) AXIS(17)
-1.0000
```

Note that some parameters in the drive can be successfully written to but the drive will not update them due to the drive functionality.

Tip: Writing zero to parameter 128 stores the drive parameters into EPROM

It is possible to perform other Minas A4N drive commands if you have a knowledge of the drive interface. Request from Panasonic the document “Network Interface Specifications” SX-DSV01601. Please note that Trio cannot supply this document.

WRITE:	READ:
DRIVE_CONTROL	DRIVE_STATUS
Command Position	Actual Position
DRIVE_INDEX	DRIVE_PARAMETER
DRIVE_SET_VAL	DRIVE_VALUE

The 3 32 bit axis parameters DRIVE_CONTROL, DRIVE_INDEX and DRIVE_SET_VAL are written cyclically by the MC464 to the drive and can be overwritten by a user program. The DRIVE_STATUS, DRIVE_PARAMETER and DRIVE_VALUE axis parameters are returned cyclically and can be read.

SLOT_NUMBER is an axis parameter which returns the drive interface module slot number that an axis is connected to.

5. Registration:

The A5N drives support drive based registration. However only a single registration channel / drive is currently implemented and it is not possible to read or write parameters or perform other commands whilst the registration is running.

The registration input is the rising edge of EXT2 (pin 11 of I/O connector):

```
REGIST(3)
WAIT UNTIL MARK
PRINT REG_POS
```

Each RTE32 Panasonic Interface has 8 time based registration inputs. Unlike other registration inputs on Trio controllers each may be used with any of the RTEX32 axes.

```
REGIST(32+channel+options) AXIS(x)
```

Tip: The axis number for the REGIST command comes from the BASE or AXIS function.

channel 0..7 selects the physical input number on the RTE32 interface.

options: add 32 to select falling edge of input, add 128 to switch on 10µsec filter.

Tip: Note how this REGIST command can be used up to 8 times with the same axis provided a different "channel" is selected.

done = R_MARK(channel) returns TRUE/FALSE has registration occurred.

v = R_REGPOS(channel) returns the registration position for the channel

The R_MARK and R_REGPOS functions are specific to an interface "slot". When multiple interfaces are fitted the SLOT(x) modifier should be used.

Pinout for Registration Connector PCB ver C and above

Channel 0	Channel 4
Channel 1	Channel 5
Channel 2	Channel 6
Channel 3	Channel 7
0V	0V
0V	0V

6. Datuming A4N Drives:

A5N drives can be datumed using the built-in datuming routines of the Motion Coordinator. To datum the A4N drives to the Z mark of the encoder the DRIVE_READ command can be used:

```
'  
' Datum A4N Drive to Z mark:  
'  
ax=2' Set axis number  
oneturn=10000' Distance for one turn depends on encoder type  
  
BASE(ax)  
  
pos = DRIVE_READ($30201) AXIS(ax)  
  
UNITS=1  
SPEED=5000  
ACC(50000)  
WDOG=ON  
  
IF pos <> -1 THEN  
    PRINT "Mechanical offset: ";pos[0]  
ELSE  
    PRINT "Drive has not yet seen Z mark"  
    MOVE(oneturn)  
    WAIT UNTIL DRIVE_READ($30201)<>-1  
    CANCEL  
    WAIT IDLE  
    pos = DRIVE_READ($30201) AXIS(ax)  
    PRINT "Mechanical offset: ";pos[0]  
ENDIF  
  
DEFPOS(pos)  
WAIT UNTIL OFFPOS=0  
MOVEABS(0)
```

7. Reset Drive Interface:

The whole drive interface can be restarted with the DRIVE_INTERFACE command:

```
DRIVE_INTERFACE(slot,0)
```